
 1 Введение

При использовании АРМ ДПУ и БД «Юпитер» рано или поздно встает вопрос организации
резервирования состояния системы на случай отказов. 

В данном документе дается описание механизма резервирования и создания аварийной копии, а
также восстановления из них работоспособного пульта «Юпитер».

 2 Общие принципы

Резервирование  АРМ  ДПУ  «Юпитер»  7.x  строится  на  принципе  создания  полной  копии
каталога  программы  с  последующим  архивированием  для  создания  серии  «снимков»  с  целью
предоставления возможности восстановления при пропуске момента возникновения ошибки. 

Для резервирования БД используется встроенный механизм создания резервной копии базы из
которой  она  может  быть  полностью  восстановлена.  Инфраструктуру  БД  (СУБД   и
специализированные утилиты редактора и администратора) потребуется установить предварительно.

Одновременно  создаются  полные  архивные  копии  программы,  которые  могут  быть
использованы для восстановления состояния системы в случае если сбой в конфигурации не был
оперативно зафиксирован и был перемещен в резервную копию.

Также  рекомендуется  хранить  полный  пакет  дистрибутивов  текущего  установленного
комплекта ПО.

Данный способ позволяет иметь полный  комплект ПО,  готовый к  восстановлению работы
пульта в случае отказа основного компьютера.

 3 Механизм

Для  создания  резервной  копии  каталога  АРМ  ДПУ,  а  также  архивов  используются
подготовленный пакетный файл.

Запуск файла может производится как в «ручном» режиме (путем создания для него ярлыка на
рабочем столе), так и по расписанию (за счет использования системного планировщика задач).

Постановка  пакетного  файла  с  планировщик  производится  автоматически  самим  пакетным
файлом при его выполнении с правами администратора (для Windows 7)

В процессе выполнения  пакетного файла производится:
1) создание каталога хранения резервной копии программы (при её отсутствии)

2) обновление резервной копии каталога программы
3) архивирование резервной копии в файл, содержащий данные о дате и времени своего создания.

4) копирование  созданного  архива  в  сетевой  каталог  хранения  (совместно  с  созданной
независимо резервной копией базы)

5) удаление «старых» копий в сетевом каталоге

6) перемещение  созданного  архива  в  каталог  хранения  архивов  (совместно  с  созданной
независимо резервной копии базы)

7) удаление «старых» копий в каталоге хранения архивов 

8) создание задания для запуска пакетного файла по расписанию в системном планировщике

Создание резервной копии БД производится специализированной утилитой, входящей в состав
программ  АРМ  БД.  В  настройках  утилиты   задается,  в  качестве  дополнительно  каталога  для
копирования созданной резервной копии, специальный каталог, используемого пакетным файлом для
создания архива АРМ ДПУ.

В случае аварии созданный архив или резервная копия содержит все необходимые настройки и
данные для начала работы на любом рабочем месте.

 4 Настройка резервирования 

Настройка резервирования включает конфигурирование параметров создания резервов БД (если
используется БД Fireird), а также развертывание и настройку пакета резервирования АРМ ДПУ.

 4.1 Развертывание и настройка пакетного файла резервирования АРМ ДПУ

Подготовленный пакет резервирования содержится в  архиве  BackUp.3.2.zip,  ссылка на  который
имеется на сайте www.elesta.ru в разделе скачивания на странице описания АРМ ДПУ.

Архив включает:

• подготовленное минимальное дерево каталогов

• исполняемый файл архиватора

• пакетный файл

• два служебных файла, задающих список исключений копирования и архивации.

Текст  пакетного  файла  (с  комментариями),  а  также  служебных  файлов  списков  исключений
приводятся в приложениях 1-3.

Пакетный файл рассчитан на размещение в каталоге  BackUp\Bin, помещаемом в корень любого
диска на рабочем месте.

Рекомендуется размещать каталог резервирования на диск отличный от основного диска работы
программы. По умолчанию пакетный файл настроен на размещение на диске E:.

Также в пакетном файле могут быть настроены следующие параметры:
• Каталог размещения программы АРМ ДПУ. По умолчанию задан каталог D:\Jupiter-16.win

• Время создания резервной копии. По умолчанию происходит в 1:10 (через 1 час после выполнения
создания резервного образа базы данных — см.п.п.4.2)

• Число архивных копий, хранящихся в основном архивном каталоге (в качестве такого каталога
используется подкаталог BackUp\Arc). По умолчанию сохраняется 31 копия архива каталога АРМ
ДПУ и 31-а резервная копия БД. Все «старые» копии стираются по мере создания новых.

• Каталог  размещения  сетевого  хранилища.  По  умолчанию  каталог  закомментирован  (в  начале
строки содержится ключ  rem)  и не используется.  Если каталог раскомментирован (ключ  rem в
начале строки удален), то производится дополнительное копирование архивов в него.

• Число архивных копий хранящихся в сетевом хранилище.  По умолчанию сохраняется 7 копий
архива каталога АРМ ДПУ и 7 резервных копий БД. Все «лишние» копии стираются по мере
создания новых.
В  качестве  сетевого  хранилища  может  использоваться  как  реальный  сетевой  ресурс,

подключенный как отдельный диск, так и установленный в компьютер FLASH-накопитель, используемый
как «чёрный ящик» на случай отказа компьютера пульта.

Рекомендуется заранее скопировать на него весь комплект дистрибутивов текущего используемого
релиза ПО.

Порядок действий по установке пакета резервирования АРМ ДПУ:
1) распаковать архив в корневой каталог диска (желательно отличного от рабочего диска АРМ ДПУ)

2) если  диск  распаковки  архива  отличается  от  E: или  каталог  АРМ  ДПУ  не  совпадает  с
D:\Jupiter  -  16.win, то внести исправления в файл save_dpu.bat:

• исправить переменную dpu_backup, задающую базовый каталог резервирования

• исправить переменную dpu_src, задающую рабочий каталог размещения АРМ ДПУ

3) запустить пакетный файл save_dpu.bat (от имени администратора для Windows7 )

4 ПО СПИ «Юпитер». Резервирование АРМ ДПУ и БД.

 4.2 Настройка резервирования БД

Настройка  резервирования  БД  включает  задание  нужного  режима  работы  сервера  БД
«Юпитер» FireBird, а также задание каталога вторичных серверов для копирования резервной копии.

Для  выполнения  резервирования  БД  необходимо  задать  режим  работы  резервирования  как
«Главный+Резервный».

Для версий БД до 4.11 это настраивается в диалоге «Настройки резервного копирования» меню
«Сервис»  приложения Администратор  БД (см.рис.4.2.1).  Следует указать  каталог  BackUp\Last как
один из каталогов вторичных серверов. Время выполнения резервного копирования по умолчанию
задается на 0:10, то есть на 1 час раньше времени выполнения пакетного файла (см.п.п.4.1)

Рис.4.2.1 Настройка параметров работы резервного копирования

Для  версии  БД  4.12  и  выше  настройка  параметров  резервного  копирования  производится
специализированной  утилитой  JupBackupConfig.exe  (см.  рис.4.2.2),  располагающейся  в  каталоге
размещения  исполняемых  файлов  БД  «Юпитер»  FireBird  (как  правило  JupDB\4\Bin).  Диалог
настройки параметров тот же (см. рис.4.2.1).

Рис.4.2.2 Программа настройки резервного копирования БД версии 4.12 и выше

 5 Запуск и проверка работы механизма резервного копирования

Для проверки работы системы резервного копирования следует:
1) убедится в наличии в списке задач системного планировщика заданий 

(для  WindowsXP:  «Панель  управления/Назначенные  задания»,  для  Windows7  и  выше
«Панель управления/Администрирование/Планировщик заданий/Библиотека планировщика»):
◦ JupBackup (создает резервную копию БД) 

◦ save_dpu (выполняет пакетный файл резервирования)

2) проверить что время запуска задач отличается на 1 час (сначала создается резервная копия БД,
затем выполняется пакетный файл)

3) проверить работоспособность заданий, запустив их из планировщика через пункт  «Выполнить»
контекстного меню соответствующего задания. 
Примечание.  При  запуске  из  планировщика  задания  выполняются  без  открытия  каких-либо
заметных окон, только состояние задания, на время выполнения, меняется на «Работает».

4) проверить  «появление»  в  каталоге  BackUp\Arc,  а  также  каталоге  сетевого  хранилища  (если
настроено), пары файлов резервной копии системы: 
◦ архив каталога АРМ ДПУ

◦ файл с расширением FBK, содержащий резервную копию БД.

В случае если при запуске из планировщика не происходит копирование резервных файлов на
сетевой  диск  необходимо,  в  настройках  задачи  в  планировщике,  изменить  имя  пользователя  на
пользователя,имеющего права доступа к сетевому диску

Примечание: учетная запись в планировщике должна иметь не пустой пароль.

Для надежность следует убедится, по прошествии суток, что в назначенное время очередная пара
файлов будет создана автоматически.

6 ПО СПИ «Юпитер». Резервирование АРМ ДПУ и БД.

 6 Восстановление пульта из резервной копии 

Для  восстановления  работы  пульта  после  отказа  компьютера  сервера  из  сохраненных
резервных  копий  (полученных  либо  с  сетевого  хранилища,  либо  сохраненных  «в  ручную»  из
основного каталога BackUp\Arc) требуется выполнить следующие действия:

1) развернуть инфраструктуру БД «Юпитер» FireBird
◦ установить СУБД FireBird из дистрибутива

◦ установить ODBC драйвера СУБД FireBird из дистрибутива

◦ установить пакет АРМ БД «Юпитер» FireBird, используя дистрибутив полной установки БД
той же версии, что использовалась на отказавшем рабочем месте.

2) поместить сохраненный файл FBK (резервная копия БД) в каталог JupDB\4\FBK развернутой
БД «Юпитер» FireBird

3) восстановить БД из резервной копии выполнив одно из следующих действий:
◦ для БД «Юпитер» FireBird, до версии 4.11 включительно

▪ запустить «Администратор БД» (для Windows7 в режиме администратора)

▪ открыть пункт меню «Сервис/Настройка резервного копирования»

▪ установить «Роль главного сервера» как «Вторичный»

▪ после закрытия диалога выполнить пункт «Сервис/Запустить резервное копирование»

▪ дождаться окончания выполнения резервного копирования (восстановления базы)

▪ открыть пункт меню «Сервис/Настройка резервного копирования»

▪ установить «Роль главного сервера» как «Главный+Резервный»

◦ для БД «Юпитер» FireBird, начиная с версии 4.12

▪ запустить «JupBackupConfig.exe»

▪ выполнить пункт  «Восстановить БД из резервной копии»,  указав в качестве файла
резервной копии файл ранее помещенный в каталог FBK.

▪ настроить  параметры  работы  резервного  копирования  установив  «Роль  главного
сервера» как «Главный+Резервный»

4) выполнить  перенос  списка  пользователей  FireBird  в  СУБД,  выполнив  в  программе
«Администраторе  БД» операцию  меню: «Пользователи/Смена  пароля  SYSDBA»,  введя
текущий пароль SYSDBA.

5) распаковать  архив  АРМ  ДПУ  (rar-файл  резервной  копии)  на  тот  же  диск  на  котором
размещался рабочий каталог АРМ ДПУ на отказавшем рабочем месте.

6) после присвоения рабочему месту тех же сетевых параметров (имя и IP-адрес) что были у
отказавшего  компьютера,  запустить  АРМ  ДПУ  (файл  jupiter.exe)  из  рабочего  каталога
программы

7) после  входа  в  программу  пульта  в  режиме  инженера  -  восстановить  список  последних
сообщений,  отображаемых  в  карточках,  выполнив  пункт  «Утилиты/Сформировать  базу
последних  сообщений» (см.рис.6.1).  Обычно  достаточно  выбрать  для  наполнения  базы
последние 3 месяца.

Рис.6.1 Настройка периода формирования базы последних сообщений

ПО СПИ «Юпитер». Резервирование АРМ ДПУ и БД. 7

Приложение 1. Текст пакетного файла архивации АРМ ДПУ (save_dpu.bat).
ECHO OFF   

rem ==========================================================

rem версия пакетного файла резервирования для инсталлятора

rem version=3.2

rem ==========================================================

rem !  каталог сетевого хранилища (при наличии)

rem SET dpu_net="E:\JUPITER"

rem ----------------------------------------------------------

rem !  путь расположения рабочего АРМ ДПУ

SET dpu_src=d:\Jupiter-16.win

rem ----------------------------------------------------------

rem !  базовый каталог архивирования

SET dpu_backup=E:\BackUp

rem ----------------------------------------------------------

rem !  время выполнения архивирования

SET dpu_time=01:10:00

rem !  число хранящихся архивных копий, 0 - число не ограничено

SET dpu_depth=31

rem !  число хранящихся архивных копий в сетевом хранилище, 0 - число не ограничено

SET dpu_depth_net=7

rem ----------------------------------------------------------

rem !  путь расположения дубликата папки АРМ ДПУ

SET dpu_dst=%dpu_backup%\Jupiter-16.win

rem ----------------------------------------------------------

rem !  каталог размещения последней копии базы и архива АРМ ДПУ

SET dpu_last=%dpu_backup%\Last

rem ----------------------------------------------------------

rem !  каталог размещения последней копии базы и архива АРМ ДПУ

SET dpu_temp=%dpu_backup%\Temp

rem ----------------------------------------------------------

rem !  каталог размещения архивов 

SET dpu_arc=%dpu_backup%\Arc

rem ----------------------------------------------------------

rem !  каталог размещения логов

SET dpu_log=%dpu_backup%\Log

rem ----------------------------------------------------------

rem технические переменные представления времени и даты

set mytime=%TIME: =0%

set hhh=%MYTIME:~0,2%

set mmm=%MYTIME:~3,2%

set sss=%MYTIME:~6,2%

set mydate=%DATE: =0%

set dd=%mydate:~0,2%

set mm=%mydate:~3,2%

set yyyy=%mydate:~6,4%

set curdate=(%yyyy%.%mm%.%dd%-%hhh%.%mmm%.%SSS%)

rem ----------------------------------------------------------

REM ! Архивация АРМ ДПУ 'Юпитер'

rem ----------------------------------------------------------

REM ! 1) создание каталогов

CD  /d %dpu_backup%"\Bin"

MKDIR %dpu_dst%

MKDIR %dpu_last%

MKDIR %dpu_temp%

MKDIR %dpu_arc%

MKDIR %dpu_log%

rem ----------------------------------------------------------

rem ==========================================================

REM ! 2) синхронизация резервной копии АРМ ДПУ

XCOPY /e /d /I /R /Y /EXCLUDE:excpydpu.lst %dpu_src% %dpu_dst%

del  %dpu_dst%\Data\sql-config\*.db3

%dpu_src%\AllSQL-backup.exe %dpu_dst%\Data\sql-config

ren  %dpu_dst%\Data\sql-config\sql-config*.db3 sql-config.db3

ren  %dpu_dst%\Data\sql-config\user_keys.*.db3 user_keys.db3

REM ----------------------------------------------------------

REM ! 3) создание архива АРМ ДПУ из резервной копии

7z.exe a -bd -r -tzip -mx9 -x@exrardpu.lst %dpu_last%\jupiter-16."win%curdate%.zip" %dpu_dst%

rem ----------------------------------------------------------

REM ! 4) копирование последних архивов БД и ДПУ в сетевое хранилище (при наличии)

IF NOT DEFINED dpu_net GOTO  move_to_arc

XCOPY /e /d /I /R /Y %dpu_last% %dpu_net%

rem ----------------------------------------------------------

:move_to_arc

rem ----------------------------------------------------------

REM ! 5) перенос последних архивов БД и ДПУ в каталог архивов

MOVE  /Y %dpu_last%\*.* %dpu_arc%

rem ----------------------------------------------------------

REM ! 6) удаление старых архивов (если задано число последних архивов и оно превышено)

IF %dpu_depth%==0 GOTO  clear_net

cd /d %dpu_arc%

for /f "skip=%dpu_depth%" %%i in ('dir /o:-d /a:-d /b jupiter*.zip') do del /Q "%%i"

for /f "skip=%dpu_depth%" %%i in ('dir /o:-d /a:-d /b *.fbk') do del /Q "%%i"

rem ----------------------------------------------------------

:clear_net

rem ----------------------------------------------------------

REM ! 7) удаление старых архивов сетевое хранилище (при наличии)

IF NOT DEFINED dpu_net GOTO  set_to_shed

IF %dpu_depth_net%==0 GOTO  set_to_shed

cd /d %dpu_net%

for /f "skip=%dpu_depth_net%" %%i in ('dir /o:-d /a:-d /b jupiter*.zip') do del /Q "%%i"

for /f "skip=%dpu_depth_net%" %%i in ('dir /o:-d /a:-d /b *.fbk') do del /Q "%%i"

rem ----------------------------------------------------------

:set_to_shed

rem ----------------------------------------------------------

REM ! 8) установка в расписание задания на новое выполнение

SET Version=Unknown

VER | FINDSTR /IL "5.0" > NUL

IF %ERRORLEVEL% EQU 0 SET Version="Windows 2000"

VER | FINDSTR /IL "5.1." > NUL

IF %ERRORLEVEL% EQU 0 SET Version="Windows XP"

VER | FINDSTR /IL "5.2." > NUL

IF %ERRORLEVEL% EQU 0 SET Version="Windows 2003"

VER | FINDSTR /IL "6.0." > NUL

IF %ERRORLEVEL% EQU 0 SET Version="Windows Vista"

VER | FINDSTR /IL "6.1." > NUL

IF %ERRORLEVEL% EQU 0 SET Version="Windows 7"

VER | FINDSTR /IL "6.2." > NUL

IF %ERRORLEVEL% EQU 0 SET Version="Windows 8"

VER | FINDSTR /IL "6.3." > NUL

IF %ERRORLEVEL% EQU 0 SET Version="Windows 8.1"

IF %VERSION% == "Windows XP" GOTO XP

IF %VERSION% == "Windows 2000" GOTO XP

IF %VERSION% == "Windows 2003" GOTO XP

IF %VERSION% == "Windows Vista" GOTO WIN7

IF %VERSION% == "Windows 7" GOTO WIN7

IF %VERSION% == "Windows 8" GOTO WIN7

IF %VERSION% == "Windows 8.1" GOTO WIN7

:XP

SCHTASKS /Delete  /TN "save_dpu" /F

SCHTASKS /Create /TN "save_dpu" /TR %dpu_backup%\Bin\save_dpu.bat /SC DAILY /ST %dpu_time% /RU SYSTEM

goto end

:WIN7

SCHTASKS /Delete  /TN "save_dpu" /F

SCHTASKS /F /Create /TN "save_dpu" /TR %dpu_backup%\Bin\save_dpu.bat /SC DAILY /ST %dpu_time% /RU SYSTEM

goto end

rem ----------------------------------------------------------

:end

10 ПО СПИ «Юпитер». Резервирование АРМ ДПУ и БД.



ООО «Элеста»                                                                                                                             СПИ «Юпитер»

Приложение 2. Файл списка исключений копирования (excpydpu.lst)

Перечислены  строки,  входящие  в  названия  файлов,  которые  не  нужно  синхронизировать  с
резервной копией  каталога  АРМ ДПУ.  Данные файлы либо  блокируются  программой в  процессе
работы, либо не требуют сохранения, так как не содержат критической информации, а место занимать
будут.

lastmsg

dex

LOG

pdf

doc

backup

rar

tmp

bak

temp

db3

Приложение 3. Файл списка исключений архивирования (exrardpu.lst)

Перечислены маски имен файлов которые не требуется включать в архив. 
Файлы либо не содержат критической информации, либо занимают необоснованно много места,

либо их содержание восстанавливается по другим файлам.

Текущий список предусматривает сохранение в архивах архива сообщений начиная с 2016 года (в
дальнейшем вы можете ввести в файл новые строки для исключения из архивирования данныъ 2017, 2018
и далее годы).

Архив обработки тревог на АРМ ДПУ, отладки и логи не сохраняются.
*\LOG\*.*

*\DATA\arm\Debug\*.*

*\DATA\arm\Alarm\*.*

*\DATA\arm\Message\200*.*

*\DATA\arm\Message\2010.*

*\DATA\arm\Message\2011.*

*\DATA\arm\Message\2012.*

*\DATA\arm\Message\2013.*

*\DATA\arm\Message\2014.*

*\DATA\arm\Message\2015.*

*\DATA\arm\LastMsg\*.*

*.pdf

*.doc

